
13 Sampling Plans

For many optimization problems, function evaluations can be quite expensive.
For example, evaluating a hardware design may require a lengthy fabrication
process, an aircraft design may require a wind tunnel test, and new deep learning
hyperparameters may require a week of GPU training. A common approach for
optimizing in contexts where evaluating design points is expensive is to build a
surrogate model, which is amodel of the optimization problem that can be efficiently
optimized in lieu of the true objective function. Further evaluations of the true
objective function can be used to improve the model. Fitting such models requires
an initial set of points, ideally points that are space-filling; that is, points that cover
the region as well as possible. This chapter covers different sampling plans for
covering the search space when we have limited resources.1 1 There are other references that

discuss the topics in this chapter
in greater detail. See, for example:
G. E. P. Box, W.G. Hunter, and J. S.
Hunter, Statistics for Experimenters:
An Introduction to Design, Data Anal-
ysis, andModel Building, 2nd ed.Wi-
ley, 2005. A. Dean, D. Voss, and
D. Draguljić, Design and Analysis
of Experiments, 2nd ed. Springer,
2017. D.C. Montgomery, Design
and Analysis of Experiments. Wiley,
2017.

13.1 Full Factorial

The full factorial sampling plan (algorithm 13.1) places a grid of evenly spaced
points over the search space. This approach is easy to implement, does not rely
on randomness, and covers the space, but it uses a large number of points. A grid
of evenly spaced points is spread over the search space as shown in figure 13.1.
Optimization over the points in a full factorial sampling plan is referred to as grid
search.

The sampling grid is defined by a lower-bound vector a and an upper-bound
vector b such that ai ≤ xi ≤ bi for each component i. For a grid with mi

samples in the ith dimension, the nearest points are separated by a distance
(bi − ai)/(mi − 1).

The full factorial method requires a sample count exponential in the number
of dimensions.2 For n dimensions with m samples per dimension, we have mn

2 The full factorial method gets its
name not from a factorial sample
count (it is exponential) but from
designingwith two ormore discrete
factors. Here the factors are the m
discretized levels associated with
each variable.

Figure 13.1. Full factorial search
covers the search space in a grid of
points.

total samples. This exponential growth is far too high to be of practical use when
there are more than a few variables. Even when full factorial sampling is able to
be used, the grid points are generally forced to be quite coarse and therefore can
easily miss small, local features of the optimization landscape.

function samples_full_factorial(a, b, m)
ranges = [range(a[i], stop=b[i], length=m[i])

for i in 1 : length(a)]
collect.(collect(Iterators.product(ranges...)))

end

Algorithm 13.1. A function for ob-
taining all sample locations for the
full factorial grid. Here, a is a vec-
tor of variable lower bounds, b is
a vector of variable upper bounds,
and m is a vector of sample counts
for each dimension.

13.2 Random Sampling

A straightforward alternative to full factorial sampling is random sampling, which
simply draws m random samples over the design space using a pseudorandom
number generator. To generate a random sample x, we can sample each variable
independently from a distribution. If we have bounds on the variables, such as
ai ≤ xi ≤ bi, a common approach is to use a uniform distribution over [ai, bi],
although other distributions may be used. For some variables, it may make sense
to use a log-uniform distribution.3 The samples of design points are uncorrelated 3 Some parameters, such as the

learning rate for deep neural net-
works, are best searched in log-
space.

with each other. The hope is that the randomness, on its own, will result in an
adequate cover of the design space.

13.3 Uniform Projection Plans

Suppose we have a two-dimensional optimization problem discretized into an
m×m sampling grid as with the full factorial method, but, instead of taking all
m2 samples, we want to sample only m positions. We could choose the samples
at random, but not all arrangements are equally useful. We want the samples to
be spread across the space, and we want the samples to be spread across each
individual component.

too clustered
no variation in
one component uniform projection

Figure 13.2. Several ways to choose
m samples from a two-dimensional
grid. We generally prefer sampling
plans that cover the space and vary
across each component.

A uniform projection plan is a sampling plan over a discrete grid where the dis-
tribution over each dimension is uniform. For example, in the rightmost sampling
plan in figure 13.2, each row has exactly one entry and each column has exactly
one entry.

A uniform projection planwith m samples on an m×m grid can be constructed
using an m-element permutation as shown in figure 13.3. There are therefore m!

possible uniform projection plans.4

4 For m = 5 this is already 5! = 120
possible plans. For m = 10 there
are 3,628,800 plans.

1 2 3 4 5

1

2

3

4

5

p = 4 2 1 3 5

Figure 13.3. Constructing a uni-
form projection plan using a per-
mutation.

Sampling with uniform projection plans is sometimes called Latin-hypercube
sampling because of the connection to Latin squares (figure 13.4). A Latin square
is an m× m grid where each row contains each integer 1 through m and each
column contains each integer 1 through m. Latin-hypercubes are a generalization
to any number of dimensions.

2 3 4 1

3 2 1 4

1 4 2 3

4 1 3 2

Figure 13.4. A 4× 4 Latin square.
A uniform projection plan can be
constructed by choosing a value
i ∈ {1, 2, 3, 4} and sampling all
cells with that value.

Uniform projection plans for n dimensions can be constructed using a permu-
tation for each dimension (algorithm 13.2).

function uniform_projection_plan(m, n)
perms = [randperm(m) for i in 1 : n]
[[perms[i][j] for i in 1 : n] for j in 1 : m]

end

Algorithm 13.2. A function for con-
structing a uniform projection plan
for an n-dimensional hypercube
with m samples per dimension. It
returns a vector of index vectors.

13.4 Stratified Sampling

Many sampling plans, including uniform projection and full factorial plans, are
based on an m×m grid. Such a grid, even if fully sampled, could miss important
information due to systematic regularities as shown in figure 13.5. One method
for providing an opportunity to hit every point is to use stratified sampling.

Stratified sampling modifies any grid-based sampling plan, including full
factorial and uniform projection plans. Cells are sampled at a point chosen uni-
formly at random from within the cell rather than at the cell’s center as shown in
figure 13.6. x

y
f (x)

sampling on grid
stratified sampling

Figure 13.5. Using an evenly-
spaced grid on a function with sys-
tematic regularities can miss im-
portant information.

Figure 13.6. Stratified sampling ap-
plied to a uniform projection plan.

13.5 Space-Filling Metrics

A good sampling plan fills the design space since the ability for a surrogate model
to generalize from samples decays with the distance from those samples. Not
all plans, even uniform projection plans, are equally good at covering the search
space. For example, a grid diagonal (figure 13.7) is a uniform projection plan but
only covers a narrow strip. This section discusses different space-filling metrics for
measuring the degree to which a sampling plan X ⊆ X fills the design space.

Figure 13.7. A uniform projection
plan that is not space-filling.

13.5.1 Discrepancy
The ability of the sampling plan to fill a hyper-rectangular design space can be
measured by its discrepancy.5 If X has low discrepancy, then a randomly chosen

5 L. Kuipers and H. Niederreiter,
Uniform Distribution of Sequences.
Dover, 2012.

subset of the design space should contain a fraction of samples proportional to the
subset’s volume.6 The discrepancy associated with X is the maximum difference

6 In arbitrary dimensions, we can
use the Lebesgue measure, which
is a generalization of volume to
any subset of n-dimensional Eu-
clidean space. It is length in one-
dimensional space, area in two-
dimensional space, and volume in
three-dimensional space.

between the fraction of samples in a hyper-rectangular subsetH and that subset’s
volume:

d(X) = supremum
H

∣
∣
∣
∣

#(X
⋂H)

#X
− λ(H)

∣
∣
∣
∣

(13.1)

where #X and #(X
⋂H) are the number of points in X and the number of points

in X that lie in H, respectively. The value λ(H) is the n-dimensional volume
of H, the product of the side lengths of H. The term supremum is very similar
to maximization but allows a solution to exist for problems where H merely
approaches a particular rectangular subset, as seen in example 13.1.7 7 The definition of discrepancy re-

quires hyper-rectangles and typi-
cally assumes that X is a finite sub-
set of a unit hypercube. The notion
of discrepancy can be extended to
allowH to include other sets, such
as convex polytopes.

Computing the discrepancy of a sampling plan over the unit hyper-rectangle
is often difficult, and it is not always clear how we can compute the discrepancy
for nonrectangular feasible sets.

13.5.2 Pairwise Distances
An alternative method determining which of two m-point sampling plans is more
space-filling is to compare the pairwise distances between all points within each
sampling plan. Sampling plans that are more spread out will tend to have larger
pairwise distances.

The comparison is typically done by sorting each set’s pairwise distances in
ascending order. The plan with the first pairwise distance that exceeds the other
is considered more space-filling.

Consider the set:

The discrepancy of X with respect to the unit square is determined by a
rectangular subsetH that either has very small area but contains very many
points or has very large area and contains very few points.

The blue rectangle, x1 ∈
[

1
10 , 2

5

]

, x2 ∈
[

1
5 , 3

5

]

, has a volume of 0.12 and
contains 3 points. Its corresponding discrepancy measure is thus 0.38.

The purple rectangle, x1 ∈
[

1
10 + ǫ, 9

10 − ǫ
]

, x2 ∈
[

1
5 + ǫ, 4

5 − ǫ
]

, produces
an even higher discrepancy. As ǫ approaches zero, the volume and the dis-
crepancy approach 0.48 because the rectangle contains no points. Note that
the limit was required, reflecting the need to use a supremum in the definition
of discrepancy.

Example 13.1. Computing the dis-
crepancy for a sampling plan over
the unit square. The sizes of the
rectangles are slightly exaggerated
to clearly show which points they
contain.

Algorithm 13.3 computes all pairwise distances between points in a sampling
plan. Algorithm 13.4 compares how well two sampling plans fill space using their
respective pairwise distances.

import LinearAlgebra: norm
function pairwise_distances(X, p=2)

m = length(X)
[norm(X[i]-X[j], p) for i in 1:(m-1) for j in (i+1):m]

end

Algorithm 13.3. A function for
obtaining the list of pairwise dis-
tances between points in sampling
plan X using the Lp norm specified
by p.

function compare_sampling_plans(A, B, p=2)
pA = sort(pairwise_distances(A, p))
pB = sort(pairwise_distances(B, p))
for (dA, dB) in zip(pA, pB)

if dA < dB
return 1

elseif dA > dB
return -1

end
end
return 0

end

Algorithm 13.4. A function for
comparing the degree towhich two
sampling plans A and B are space-
filling using the Lp norm specified
by p. The function returns -1 if A is
more space-filling than B. It returns
1 if B is more space-filling than A. It
returns 0 if they are equivalent.

Onemethod for generating a space-filling uniformprojection plan is to generate
several candidates at random and then use the one that is most space-filling.

We can search for a space-filling uniform projection plan by repeatedly mu-
tating a uniform projection plan in a way that preserves the uniform projection
property (algorithm 13.5). Simulated annealing, for example, could be used to
search the space for a sampling plan with good coverage.

function mutate!(X)
m, n = length(X), length(X[1])
j = rand(1:n)
i = randperm(m)[1:2]
X[i[1]][j], X[i[2]][j] = X[i[2]][j], X[i[1]][j]
return X

end

Algorithm 13.5. A function for mu-
tating uniform projection plan X,
while maintaining its uniform pro-
jection property.

13.5.3 Morris-Mitchell Criterion
The comparison scheme in section 13.5.2 typically results in a challenging opti-
mization problem with many local minima. An alternative is to optimize with
respect to the Morris-Mitchell criterion (algorithm 13.6):8 8 M.D. Morris and T. J. Mitchell,

“Exploratory Designs for Compu-
tational Experiments,” Journal of
Statistical Planning and Inference,
vol. 43, no. 3, pp. 381–402, 1995.Φq(X) =

(

∑
i

d
−q
i

)1/q

(13.2)

where di is the ith pairwise distance between points in X and q > 0 is a tunable
parameter.9 Morris and Mitchell recommend optimizing: 9 Larger values of q will give higher

penalties to large distances.
minimize

X
maximize

q∈{1,2,5,10,20,50,100}
Φq(X) (13.3)

function phiq(X, q=1, p=2)
dists = pairwise_distances(X, p)
return sum(dists.^(-q))^(1/q)

end

Algorithm 13.6. An implemen-
tation of the Morris-Mitchell cri-
terion which takes a list of de-
sign points X, the criterion param-
eter q > 0, and a norm parameter
p ≥ 1.

Figure 13.8 shows the Morris-Mitchell criterion evaluated for several randomly
generated uniform projection plans.

Φ1 = 23.7 Φ1 = 24.2 Φ1 = 24.2 Φ1 = 24.3 Φ1 = 24.3 Φ1 = 24.3

Φ1 = 24.3 Φ1 = 24.5 Φ1 = 24.8 Φ1 = 24.9 Φ1 = 25.0 Φ1 = 25.0

Φ1 = 25.3 Φ1 = 25.5 Φ1 = 26.2 Φ1 = 27.2 Φ1 = 33.0 Φ1 = 36.7

Figure 13.8. Uniform projection
plans sorted from best to worst ac-
cording to Φ1.

13.6 Space-Filling Subsets

In some cases, we have a set of points X and want to find a subset of points
S ⊂ X that still maximally fills X. The need for identifying space-filling subsets of
X arises in the context of multifidelity models.10 For example, suppose we used a 10 A. I. J. Forrester, A. Sóbester, and

A. J. Keane, “Multi-Fidelity Op-
timization via Surrogate Mod-
elling,” Proceedings of the Royal So-
ciety of London A: Mathematical,
Physical and Engineering Sciences,
vol. 463, no. 2088, pp. 3251–3269,
2007.

sampling plan X to identify a variety of aircraft wing designs to evaluate using
computational fluid dynamic models in simulation. We can choose only a subset
of these design points S to build and test in a wind tunnel. We still want S to be
space filling.

The degree to which S fills the design space can be quantified using the max-
imum distance between a point in X and the closest point in S. This metric
generalizes to any two finite sets A and B (algorithm 13.7). We can use any Lp

norm, but we typically use L2, the Euclidean distance:

dmax(X, S) = maximize
x∈X

minimize
s∈S

‖s− x‖p (13.4)

min_dist(a, B, p) = minimum(norm(a-b, p) for b in B)
d_max(A, B, p=2) = maximum(min_dist(a, B, p) for a in A)

Algorithm 13.7. The set Lp distance
metrics between two discrete sets,
where A and B are lists of design
points and p is the Lp norm param-
eter.A space-filling sampling plan is one that minimizes this metric.11 Finding a
11 We can also minimize the Morris-
Mitchell criterion for S.space-filling sampling plan with m elements is an optimization problem

minimize
S

dmax(X, S)

subject to S ⊆ X

#S = m

(13.5)

Optimizing equation (13.5) is typically computationally intractable. A brute
force approachwould try all d!/m!(d−m)! size-m subsets for a dataset of d design
points. Both greedy local search (algorithm 13.8) and the exchange algorithm (algo-
rithm 13.9) are heuristic strategies for overcoming this difficulty. They typically
find acceptable space-filling subsets of X.

Greedy local search starts with a point selected randomly from X and incre-
mentally adds the next best point that minimizes the distance metric. Points are
added until the desired number of points is reached. Because the points are ini-
tialized randomly, the best results are obtained by running greedy local search
several times and keeping the best sampling plan (algorithm 13.10).

function greedy_local_search(X, m, d=d_max)
S = [sample(X)]
for i in 2 : m

j = argmin([x ∈ S ? Inf : d(X, push!(copy(S), x))
for x in X])

push!(S, X[j])
end
return S

end

Algorithm 13.8. Greedy local
search, for finding m-element sam-
pling plans that minimize a dis-
tance metric d for discrete set X.

The exchange algorithm initializes S to a random subset of X and repeatedly
replaces points that are in S with a different point in X that is not already in S

to improve on the distance metric. The exchange algorithm is also typically run
multiple times.

Figure 13.9 compares space-filling subsets obtained using greedy local search
and the exchange algorithm.

13.7 Quasi-Random Sequences

Quasi-random sequences,12 also called low-discrepancy sequences, are often used in 12 C. Lemieux, Monte Carlo and
Quasi-Monte Carlo Sampling.
Springer, 2009.the context of trying to approximate an integral over a multidimensional space:

∫

X
f (x) dx ≈ v

m

m

∑
i=1

f (x(i)) (13.6)

where each x(i) is sampled uniformly at random over the domain X and v is the
volume of X . This approximation is known as Monte Carlo integration.

Rather than relying on random or pseudorandom numbers to generate inte-
gration points, quasi-random sequences are deterministic sequences that fill the
space in a systematic manner so that the integral converges as fast as possible in
the number of points m.13 These quasi-Monte Carlo methods have an error conver-

13 Pseudorandom number se-
quences, such as those produced
by a sequence of calls to rand, are
deterministic given a particular
seed, but they appear random.
Quasi-random numbers are also
deterministic but do not appear
random.gence of O(1/m) as opposed to O(1/

√
m) for typical Monte Carlo integration, as

shown in figure 13.10.

function exchange_algorithm(X, m, d=d_max)
S = sample(X, m, replace=false)
δ, done = d(X, S), false
while !done

best_pair = (0,0)
for i in 1 : m

s = S[i]
for (j,x) in enumerate(X)

if !in(x, S)
S[i] = x
δ′ = d(X, S)
if δ′ < δ

δ = δ′
best_pair = (i,j)

end
end

end
S[i] = s

end
done = best_pair == (0,0)
if !done

i,j = best_pair
S[i] = X[j]

end
end
return S

end

Algorithm 13.9. The exchange al-
gorithm for finding m-element sam-
pling plans that minimize a dis-
tance metric d for discrete set X.

function multistart_local_search(X, m, alg, k_max, d=d_max)
sets = [alg(X, m, d) for i in 1 : k_max]
return sets[argmin([d(X, S) for S in sets])]

end

Algorithm 13.10. Multistart local
search runs a particular search
algorithm multiple times and re-
turns the best result. Here, X is
the list of points, m is the size
of the desired sampling plan,
alg is either exchange_algorithm
or greedy_local_search, k_max is
the number of iterations to run,
and d is the distance metric.

Figure 13.9. Space-filling sub-
sets obtained with both greedy lo-
cal search and the exchange algo-
rithm.

Figure 13.10. The error from
estimating

∫ 1
0 sin(10x) dx using

Monte Carlo integration with ran-
dom numbers from U(0, 1) and
a Sobol sequence. The Sobol se-
quence, covered in section 13.7.3,
converges faster.

Quasi-random sequences are typically constructed for the unit n-dimensional
hypercube, [0, 1]n. Any multidimensional function with bounds on each variable
can be transformed into such a hypercube. This transformation is implemented
in algorithm 7.9.

Various methods exist for generating quasi-random sequences. Several such
methods are compared to random sampling in figure 13.12.

13.7.1 Additive Recurrence
Simple recurrence relations of the form:

x(k+1) = x(k) + c (mod 1) (13.7)

produce space-filling sets provided that c is irrational. The value of c leading to
the smallest discrepancy is

c = 1− ϕ =

√
5− 1

2
≈ 0.618034 (13.8)

where ϕ is the golden ratio.14 14 C. Schretter, L. Kobbelt, and P.-O.
Dehaye, “Golden Ratio Sequences
for Low-Discrepancy Sampling,”
Journal of Graphics Tools, vol. 16,
no. 2, pp. 95–104, 2016.

We can construct a space-filling set over n dimensions using an additive recur-
rence sequence for each coordinate, each with its own value of c. The square roots
of the primes are known to be irrational, and can thus be used to obtain different
sequences for each coordinate:

c1 =
√

2, c2 =
√

3, c3 =
√

5, c4 =
√

7, c5 =
√

11, . . . (13.9)

Methods for additive recurrence are implemented in algorithm 13.11.

using Primes
function get_filling_set_additive_recurrence(m; c=φ-1)

X = [rand()]
for i in 2 : m

push!(X, mod(X[end] + c, 1))
end
return X

end
function get_filling_set_additive_recurrence(m, n)

ps = primes(max(ceil(Int, n*(log(n) + log(log(n)))), 13))
seqs = [get_filling_set_additive_recurrence(m, c=sqrt(p))

for p in ps[1:n]]
return [collect(x) for x in zip(seqs...)]

end

Algorithm 13.11. Additive
recurrence for constructing
m-element filling sequences over
n-dimensional unit hypercubes.
The Primes package is used to
generate the first n prime numbers,
where the kth prime number is
bounded by

k(log k + log log k)

for k > 6, and primes(a) returns
all primes up to a. Note that 13 is
the sixth prime number.

13.7.2 Halton Sequence
The Halton sequence is a multidimensional quasi-random space-filling set.15 The 15 J.H. Halton, “Algorithm 247:

Radical-Inverse Quasi-Random
Point Sequence,” Communications
of the ACM, vol. 7, no. 12, pp. 701–
702, 1964.

single-dimensional version, called van der Corput sequences, generates sequences
where the unit interval is divided into powers of base b. For example, b = 2

produces:
X =

{
1

2
,

1

4
,

3

4
,

1

8
,

5

8
,

3

8
,

7

8
,

1

16
, . . .

}

(13.10)

whereas b = 5 produces:

X =

{
1

5
,

2

5
,

3

5
,

4

5
,

1

25
,

6

25
,

11

25
, . . .

}

(13.11)

Multi-dimensional space-filling sequences use one van der Corput sequence for
each dimension, each with its own base b. The bases, however, must be coprime16 16 Two integers are coprime if the

only positive integer that divides
them both is 1.in order to be uncorrelated. Methods for constructing Halton sequences are

implemented in algorithm 13.12.

using Primes
function halton(i, b)

result, f = 0.0, 1.0
while i > 0

f = f / b;
result = result + f * mod(i, b)
i = floor(Int, i / b)

end
return result

end
get_filling_set_halton(m; b=2) = [halton(i,b) for i in 1: m]
function get_filling_set_halton(m, n)

bs = primes(max(ceil(Int, n*(log(n) + log(log(n)))), 6))
seqs = [get_filling_set_halton(m, b=b) for b in bs[1:n]]
return [collect(x) for x in zip(seqs...)]

end

Algorithm 13.12. Halton
quasi-random m-element filling
sequences over n-dimensional unit
hypercubes, where b is the base.
The bases bs must be coprime.

Figure 13.11. The Halton sequence
with b = [19, 23] for which the
first 18 samples are perfectly lin-
early correlated.For large primes, we can get correlation in the first few numbers. Such a corre-

lation is shown in figure 13.11. Correlation can be avoided by the leaped Halton
method,17 which takes every pth point, where p is a prime different from all 17 L. Kocis and W. J. Whiten,

“Computational Investigations
of Low-Discrepancy Sequences,”
ACM Transactions on Mathematical
Software, vol. 23, no. 2, pp. 266–294,
1997.

coordinate bases.

13.7.3 Sobol Sequences
Sobol sequences are quasi-random space-filling sequences for n-dimensional hy-
percubes.18 They are generated by xor-ing the previous Sobol number with a set 18 I.M. Sobol, “On the Distribution

of Points in a Cube and the Ap-
proximate Evaluation of Integrals,”
USSR Computational Mathematics
and Mathematical Physics, vol. 7,
no. 4, pp. 86–112, 1967.

of direction numbers:19

19 The symbol ⊻ denotes the xor op-
eration, which returns true if and
only if both inputs are different.

X
(i)
j = X

(i−1)
j ⊻ v

(k)
j (13.12)

where v
(k)
j is the jth bit of the kth direction number. Tables of good direction

numbers have been provided by various authors.20

20 The Sobol.jl package provides
an implementation for up to 1,111
dimensions.

A comparison of these and previous approaches is shown in figure 13.12. For
high values several methods exhibit a clear underlying structure.

Figure 13.12. A comparison of
space-filling sampling plans in two
dimensions. Samples are colored
according to the order in which
they are sampled. The uniform
projection plan was generated ran-
domly and is not optimized.

13.8 Summary

• Sampling plans are used to cover search spaces with a limited number of
points.

• Full factorial sampling, which involves sampling at the vertices of a uniformly
discretized grid, requires a number of points exponential in the number of
dimensions.

• Uniform projection plans, which project uniformly over each dimension, can
be efficiently generated and can be optimized to be space filling.

• Greedy local search and the exchange algorithm can be used to find a subset
of points that maximally fill a space.

• Quasi-random sequences are deterministic procedures by which space-filling
sampling plans can be generated.

13.9 Exercises

Exercise 13.1. Filling a multidimensional space requires exponentially more
points as the number of dimensions increases. To help build this intuition, deter-
mine the side lengths of an n-dimensional hypercube such that it fills half of the
volume of the n-dimensional unit hypercube.

Exercise 13.2. Suppose that you sample randomly inside a unit sphere in n

dimensions. Compute the probability that a randomly sampled point is within ǫ

distance from the surface of the sphere as n→ ∞. Hint: The volume of a sphere
is C(n)rn, where r is the radius and C(n) is a function of the dimension n only.
Exercise 13.3. Suppose we have a sampling plan X = {x(1), . . . , x(10)}, where

x(i) = [cos(2πi/10), sin(2πi/10)] (13.13)

Compute the Morris-Mitchell criterion for X using an L2 norm when the parame-
ter q is set to 2. In other words, evaluate Φ2(X). If we add [2, 3] to each x(i), will
Φ2(X) change? Why or why not?
Exercise 13.4. Additive recurrence requires that the multiplicative factor c in
equation (13.7) be irrational. Why can c not be rational?

